

Free Preview – Week #30
Learn Excel from Mr Excel

Encourage your friends to sign up at
http://www.mrexcel.com/learn-excel.html

This week - five topics on using super powerful CSE formulas.

CREATE A CSE FORMULA TO BUILD A SUPER FORMULA

Problem: Your dataset has a column with quantity sold and another column with unit prices, as shown in Fig. 587. You want one formula to figure out the total of quantity times unit price.

	A	B	C	D	E	F
1	**Region**	**Product**	**Date**	**Customer**	**Quantity**	**Unit Price**
553	West	XYZ	23-Dec-04	Verizon	900	95
554	East	ABC	24-Dec-04	Verizon	100	87
555	West	DEF	24-Dec-04	Ainsworth	500	78
556	Central	XYZ	24-Dec-04	Wal-Mart	200	95
557	Central	ABC	26-Dec-04	Ainsworth	100	87
558	West	DEF	26-Dec-04	Wal-Mart	700	78
559	East	XYZ	26-Dec-04	Exxon	600	95
560	West	XYZ	26-Dec-04	SBC Communicatic	500	95
561	Central	ABC	27-Dec-04	Sun Life Financial	500	87
562	Central	ABC	28-Dec-04	General Motors	600	87
563	Central	ABC	28-Dec-04	General Motors	600	87
564	Central	ABC	28-Dec-04	General Motors	900	87

Fig. 587

Strategy: The typical strategy is to add a new column with price times quantity and add that up. However, this is not necessary if you use a type of super formula that I call the CSE formula.

=SUM(E2:E564*F2:F564)

If you've been using Excel for a while, you will think that this looks like it will not work. In fact, if you enter the formula, you will get a #VALUE! error, confirming that it does not work, as shown in Fig. 588.

F566 =SUM(E2:E564*F2:F564)

	D	E	F	G	H
1	**Customer**	**Quantity**	**Unit Price**		
563	General Motors	600	87		
564	General Motors	900	87		
565					
566			#VALUE!		
567					

Fig. 588

However, if you know the secret you can still make the formula work. Edit the formula by hitting F2. Instead of using Enter to finish the formula, hold down Ctrl+Shift and then hit Enter.

CSE=Ctrl+Shift+Enter

As shown in Fig. 589, miraculously, Excel does 563 multiplications and then adds them up to give you a result.



Fig. 589

In the formula bar above, you will note that there are curly braces around the formula. You do not enter these braces. Excel adds them when you use Ctrl+Shift+Enter.

Additional Details: Most people will only have an occasion to use this formula once a month. I found that I could never remember the keystroke combination, so I renamed these formulas "CSE" to help me to remember Ctrl+Shift+Enter. If you need to search Microsoft Help on the subject, check under the name "Array Formulas".

Summary: One single CSE formula can replace hundreds or even thousands of intermediate formulas.

Commands Discussed: Control+Shift+Enter

LEARN TO USE BOOLEAN LOGIC FACTS TO SIMPLIFY LOGIC

Problem: If you have to enter multiple IF conditions, having a good understanding of logical operators will help you to simplify the formula.

Strategy: When you are dealing with conditions, the language is full of And, Or, Not, Nor, True, and False. All of these words have mathematical equivalents. Understanding them will enable you to build complex two-condition formulas.

A Boolean formula returns either TRUE or FALSE. In Fig. 590, the formula =A2>100 will return TRUE.

B2 fx =A2>100

	A	B	C
1	Sales	Sales>100?	
2	105	TRUE	
3	92	FALSE	
4	85	FALSE	
5	100	FALSE	
6	101	TRUE	

Fig. 590

You can have many such tests. As shown in Fig. 591, this dataset has columns to test if the product is a particular product line or if the region is a particular region.

E2 fx =B2="ABC"

	A	B	C	D	E	F
1	Sales	Product	Region	Sales>100?	Prod=ABC?	Region=East?
2	105	ABC	East	TRUE	TRUE	TRUE
3	92	ABC	West	FALSE	TRUE	FALSE
4	85	ABC	East	FALSE	TRUE	TRUE
5	100	DEF	West	FALSE	FALSE	FALSE
6	101	DEF	East	TRUE	FALSE	TRUE

Fig. 591

You can build a calculation from the results of multiple Boolean formulas. One popular operator in Boolean logic is the AND operator. If you want to know if D2 AND E2 is TRUE, you can state this as a formula.

In Boolean Logic,

- Think of each TRUE as the number 1.
- Think of each FALSE as the number 0.

- Think of each AND as a Multiplication Operator
- Think of each OR as an Addition Operator

If the result of the calculation is 0, then the answer is FALSE. If the result of the calculation is non-zero then the answer is TRUE.

Here is an example, as shown in Fig. 592:

Plain language: The bonus is paid if the sale is >100 and the product is ABC.

Excel: =(A2>100)*(B2="ABC")

A=105 B=ABC: TRUE * TRUE= 1*1=1=TRUE

A=92 B=ABC: FALSE * TRUE= 0*1=0=FALSE

A=85 B=DEF: FALSE * FALSE= 0*0=0=FALSE

A=101 B=DEF: TRUE * FALSE= 1*0=0=FALSE

	A	B	C	D	E	F	G
1	Sales	Product	Region	Sales>100?	Prod=ABC?	Region=East?	Bonus?
2	105	ABC	East	TRUE	TRUE	TRUE	1
3	92	ABC	West	FALSE	TRUE	FALSE	0
4	85	DEF	East	FALSE	FALSE	TRUE	0
5	101	DEF	West	TRUE	FALSE	FALSE	0

Fig. 592

Here are the logic rules for AND operators and OR operators.

AND	**OR**
TRUE*TRUE=TRUE	TRUE+TRUE=TRUE
TRUE*FALSE=FALSE	TRUE+FALSE=TRUE
FALSE*TRUE=FALSE	FALSE+TRUE=TRUE
FALSE*FALSE=FALSE	FALSE+FALSE=FALSE

Here is another example to work through.

Plain language: Bonus is paid for selling any item over $100.00 or for sales of DEF product.

Excel: =(A2>100)+(B2="DEF"), as shown in Fig. 593.

Sales	Product	Sales>100	Item=DEF	Bonus Calculation
80	DEF	FALSE	TRUE	=0+1=1=TRUE
105	DEF	TRUE	TRUE	=1+1=2=TRUE
90	ABC	FALSE	FALSE	=0+0=0=FALSE
110	ABC	TRUE	FALSE	=1+0=1=TRUE

F2 fx =(A2>100)+(B2="DEF")

	A	B	C	D	E	F
1	Sales	Product	Region	Sales>100?	Prod=DEF?	Bonus?
2	80	DEF	East	FALSE	TRUE	1
3	105	DEF	West	TRUE	TRUE	2
4	90	ABC	East	FALSE	FALSE	0
5	110	ABC	West	TRUE	FALSE	1

Fig. 593

Using the above rules, you can write complex sets of Boolean logic. The formula in Fig. 594 would pay a $25 bonus for all West region sales of jackets at any price or caps above $50.

D2 fx =IF(((B2="Jacket")+((B2="Cap")*(A2>50)))*(C2="West"),25,0)

	A	B	C	D	E	F	G	H	I
1	Price	Product	Region	Bonus?					
2	225	Jacket	East	0					
3	250	Jacket	West	25					
4	35	Cap	East	0					
5	55	Cap	West	25					
6	175	Jacket	East	0					
7	200	Jacket	West	25					
8	55	Cap	East	0					
9	35	Cap	West	0					

Fig. 594

Summary: Excel does offer the AND and OR functions. However, being able to use Boolean terms as the first parameter of an IF statement allows for more complex calculations.

Functions Discussed: =IF(); =AND(); =OR()

REPLACE IF FUNCTION WITH BOOLEAN LOGIC

Problem: As shown in Fig. 595, you need to calculate a 10 percent bonus on sales greater than $1000.00.

	A	B	C	D	E	F
1	Invoice	Product	Price	Qty	Total	Bonus
2	2010	Jacket	225	14	3150	
3	2011	Jacket	225	8	1800	
4	2012	Jacket	225	11	2475	
5	2013	Cap	25	6	150	
6	2014	Jacket	225	17	3825	

Fig. 595

Strategy: You can use the Boolean logic facts to do this calculation without an IF function. Remember that a Boolean test that results in a TRUE is treated as a 1 and a FALSE statement is treated as a 0. Thus, you could multiply the calculation E2*0.1 by the Boolean test (E2>1000), as shown in Fig. 596.

F2 fx =(E2*0.1)*(E2>1000)

	A	B	C	D	E	F
1	Invoice	Product	Price	Qty	Total	Bonus
2	2010	Jacket	225	14	3150	315
3	2011	Jacket	225	8	1800	180
4	2012	Jacket	225	11	2475	247.5
5	2013	Cap	25	6	150	0
6	2014	Jacket	225	17	3825	382.5
7	2015	Jacket	225	3	675	0
8	2016	Jacket	225	2	450	0

Fig. 596

Summary: This formula combines a math calculation with a Boolean test to produce a valid result. Using these types of calculations is one key to using conditional sums with two conditions. You will learn about these in the next chapter.

Functions Discussed: =IF()

TEST FOR TWO CONDITIONS IN A SUM

Problem: You need to sum a dataset based on two conditions. The SUMIF function can only handle one condition. As shown in Fig. 597, you want to write a formula that will total all sales by Amy of product ABC.

	A	B	C	D	E	F
1	Invoice	Rep	Customer	Product	Sales	
91	1099	Deb	L3917	XYZ	244	
92	1100	Deb	F8477	XYZ	198	
93	1101	Ben	Q7142	ABC	345	
94	1102	Chaz	D8695	DEF	342	
95	1103	Amy	J9467	ABC	245	
96	1104	Amy	M8580	XYZ	163	
97	1105	Chaz	N8977	ABC	202	
98	1106	Ben	K5508	ABC	116	
99	1107	Ben	W9236	ABC	188	
100						
101			ABC	DEF	XYZ	
102		Amy	=SUMIF(			
103		Ben				
104		Chaz				
105		Deb				

SUMIF(**range**, criteria, [sum_range])

Fig. 597

Strategy: You have to use a CSE formula to do this. This question comes up a lot at the MrExcel.com site. It is a definite collision point in Excel – it is a very common problem, but a very difficult solution.

The SUMIF Function will not do this for you. You have to use Boolean logic and a CSE formula. You want to test to see if each cell in B2:B99 is equal to B102. This would be represented by the following formula:

=($B$2:$B$99=$B102)

You also want to test to see if each cell in D2:D99 is equal to C101. This would be represented by the following formula:

=($D$2:$D$99=C$101)

If you multiply those two terms together, you will end up with a 1 wherever both conditions are True and a 0 wherever one condition is not True.

=($B$2:$B$99=$B102)*($D$2:$D$99=C$101)

As shown in Fig. 598, pretend that you actually entered these formulas in columns F, G, and H. After you have the 1 or 0 in column H, you have to multiply that result times Sales in column E and then sum up column I.

I2 =+H2*E2

	A	B	C	D	E	F	G	H	I
1	Invoice	Rep	Customer	Product	Sales	*Amy?*	*ABC?*	*F*G*	*Result*
2	1018	Chaz	A1700	ABC	320	*FALSE*	*TRUE*	*0*	0
3	1051	Amy	A3203	XYZ	309	*TRUE*	*FALSE*	*0*	0
4	1086	Amy	A3446	DEF	150	*TRUE*	*FALSE*	*0*	0
5	1049	Chaz	A4580	XYZ	228	*FALSE*	*FALSE*	*0*	0
6	1054	Chaz	A6969	ABC	260	*FALSE*	*TRUE*	*0*	0
7	1034	Amy	A7717	ABC	251	*TRUE*	*TRUE*	*1*	251
8	1077	Chaz	B5387	ABC	228	*FALSE*	*TRUE*	*0*	0

Fig. 598

To multiply the Boolean terms by Sales, use:

=($B$2:$B$99=$B102)*($D$2:$D$99=C$101)*($E$2:$E$99)

To sum the result, use:

=SUM(($B$2:$B$99=$B102)*($D$2:$D$99=C$101)*($E$2:$E$99))

Enter this formula in C102. Instead of using Enter after typing the formula, use Ctrl+Shift+Enter. Excel will evaluate the formula as an array and produce the correct result, as shown in Fig. 599.

C102 {=SUM(($B$2:$B$99=$B102)*($D$2:$D$99=C$101)*($E$2:$E$99))}

	A	B	C	D	E	F	G	H	I	J
1	Invoice	Rep	Customer	Product	Sales					
98	1020	Deb	Z2412	ABC	233					
99	1069	Amy	Z9822	DEF	278					
100										
101			ABC	DEF	XYZ					
102		Amy	998							
103		Ben								
104		Chaz								
105		Deb								

Fig. 599

Note: *You do NOT type the curly braces around the formula. Excel will add those when you use Ctrl+Shift+Enter.*

Due to the careful use of the dollar signs in each reference, you've made a formula that can be copied to the rest of the table. Normally, you would copy C102 and paste it to C102:E105, but because of the limitation of CSE formulas, you cannot do this. You first have to copy C102 to C103:C105, as shown in Fig. 600.

	A	B	C	
1	Invoice	Rep	Customer	Pro
98	1020	Deb	Z2412	AB
99	1069	Amy	Z9822	DE
100				
101			ABC	DE
102		Amy	998	
103		Ben	2210	
104		Chaz	2456	
105		Deb	1181	

Fig. 600

Then, copy C102:C105 and paste it to D102:E105, as shown in Fig. 601.

	A	B	C	D	E
1	Invoice	Rep	Customer	Product	Sales
98	1020	Deb	Z2412	ABC	233
99	1069	Amy	Z9822	DEF	278
100					
101			ABC	DEF	XYZ
102		Amy	998	2687	1723
103		Ben	2210	1551	1447
104		Chaz	2456	839	1389
105		Deb	1181	1870	2106

Fig. 601

Gotcha: CSE or array formulas are very powerful. They are also very memory intensive. Don't go overboard with them. I once tried to build a report of 800 CSE formulas with each one totaling a 50,000-row dataset using three conditions. If I hadn't rebooted the computer, it would still be trying to calculate the formula.

Alternate Strategy: Aladin Akyurek has written an excellent article about this topic at http://www.mrexcel.com/wwwboard/messages/8961.html. Aladin notes that this problem can be solved without using a CSE formula by using the SUMPRODUCT function. The equivalent formula for cell C102 would be as follows:

=SUMPRODUCT(($B$2:$B$99=$B102)*($D$2:$D$99=C$101)*($E$2:$E$99))

Summary: Using CSE formulas and Boolean logic, you can solve a problem where you need to sum on the basis of two conditions. This is not just a powerful extension of SUMIF. Using these types of formulas, you can write just about any conditional calculation that you can imagine.

CAN THE RESULTS OF A FORMULA BE USED IN COUNTIF?

Problem: You need to count the number of students who are above average. The student grades are arranged from B2:B26, as shown in Fig. 602.

	A	B
1	Name	Score
2	Allen	92
3	Barb	72
4	Bob	60
5	Carla	82
6	Charley	72
7	Ed	64
8	Gail	72
9	Katia	82

Fig. 602

Strategy: The second parameter of the COUNTIF can be a calculation. As shown in Fig. 603, you can concatenate a text operator with a calculation, such as the one that follows.

=COUNTIF(B2:B26,“>”&AVERAGE(B2:B26))

D2 =COUNTIF(B2:B26,">"&AVERAGE(B2:B26))

	A	B	C	D	E	F	G
1	Name	Score					
2	Allen	92		13			
3	Barb	72					
4	Bob	60					

Fig. 603

Summary: Using the result of a calculation as the criteria argument for the COUNTIF opens up a wide possibility of measurements using COUNTIF and SUMIF.

Functions Discussed: =SUMIF(); =COUNTIF()